

Outline of trig integration rules

1. $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \wedge m \in \mathbb{Z} \wedge n \in \mathbb{Z}$

1: $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \wedge \frac{m}{2} \in \mathbb{Z} \wedge \frac{n}{2} \in \mathbb{Z}$

2. $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \wedge \frac{m-1}{2} \in \mathbb{Z} \wedge \frac{n-1}{2} \in \mathbb{Z}$

1: $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $\frac{p}{2} \in \mathbb{Z}^- \wedge \frac{m-1}{2} \in \mathbb{Z} \wedge \frac{n-1}{2} \in \mathbb{Z}$

2: $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $\frac{p-1}{2} \in \mathbb{Z}^- \wedge \frac{m-1}{2} \in \mathbb{Z} \wedge \frac{n-1}{2} \in \mathbb{Z}$

3: $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \wedge \frac{m}{2} \in \mathbb{Z} \wedge \frac{n-1}{2} \in \mathbb{Z}$

2. $\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \wedge m \in \mathbb{Z} \wedge n \in \mathbb{Z}$

1: $\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \wedge \frac{m}{2} \in \mathbb{Z} \wedge \frac{n}{2} \in \mathbb{Z}$

2: $\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx$ when $\frac{p-1}{2} \in \mathbb{Z}^- \wedge \frac{m}{2} \in \mathbb{Z} \wedge \frac{n-1}{2} \in \mathbb{Z}$

3: $\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \wedge m \in \mathbb{Z} \wedge n \in \mathbb{Z}$

Rules for integrands of the form $(a \sin[m(c+dx)] + b \sin[n(c+dx)])^p$

1: $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$

■ **Derivation: Integration by substitution**

■ **Basis:** If $\frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$, then

$$F[\sin[m(c+dx)], \sin[n(c+dx)]] = \frac{1}{d} \operatorname{Subst}\left[\frac{F[\sin[m \operatorname{ArcTan}[x]], \sin[n \operatorname{ArcTan}[x]]]}{1+x^2}, x, \tan[c+dx]\right] \partial_x \tan[c+dx]$$

■ **Basis:** If $\frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$, then

$$F[\cos[m(c+dx)], \cos[n(c+dx)]] = -\frac{1}{d} \operatorname{Subst}\left[\frac{F[\cos[m \operatorname{ArcCot}[x]], \cos[n \operatorname{ArcCot}[x]]]}{1+x^2}, x, \cot[c+dx]\right] \partial_x \cot[c+dx]$$

■ **Note:** If $\frac{m}{2} \in \mathbb{Z}$, then $\sin[m \operatorname{ArcTan}[x]]$ equals x times a rational functions in x^2 .

■ **Note:** If $\frac{m}{2} \in \mathbb{Z}$, then $\cos[m \operatorname{ArcCot}[x]]$ equals a rational functions in x^2 .

■ **Rule:** If $p \in \mathbb{Z}^- \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$, then

$$\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx \rightarrow \frac{1}{d} \operatorname{Subst}\left[\int \frac{(a \sin[m \operatorname{ArcTan}[x]] + b \sin[n \operatorname{ArcTan}[x]])^p}{1+x^2} dx, x, \tan[c+dx]\right]$$

■ **Program code:**

```
Int[(a_.*sin[m_.*(c_.+d_.*x_)]+b_.*sin[n_.*(c_.+d_.*x_)])^p_,x_Symbol] :=
  1/d * Subst[Int[Simplify[TrigExpand[a*sin[m*ArcTan[x]]+b*sin[n*ArcTan[x]]]^p/(1+x^2),x],x,Tan[c+d*x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IntegerQ[m/2] && IntegerQ[n/2]
```

```
Int[(a_.*cos[m_.*(c_.+d_.*x_)]+b_.*cos[n_.*(c_.+d_.*x_)])^p_,x_Symbol] :=
  -1/d * Subst[Int[Simplify[TrigExpand[a*cos[m*ArcCot[x]]+b*cos[n*ArcCot[x]]]^p/(1+x^2),x],x,Cot[c+d*x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IntegerQ[m/2] && IntegerQ[n/2]
```

2. $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$

1: $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $\frac{p}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$

■ **Derivation: Integration by substitution**

■ **Basis:** If $\frac{p}{2} \in \mathbb{Z} \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$, then

$$(a \sin[m(c+dx)] + b \sin[n(c+dx)])^p = \frac{1}{d} \operatorname{Subst} \left[\frac{(a \sin[m \operatorname{ArcTan}[x]] + b \sin[n \operatorname{ArcTan}[x]])^p}{1+x^2}, x, \tan[c+dx] \right] \partial_x \tan[c+dx]$$

■ **Basis:** If $\frac{p}{2} \in \mathbb{Z} \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$, then

$$(a \cos[m(c+dx)] + b \cos[n(c+dx)])^p = \frac{1}{d} \operatorname{Subst} \left[\frac{(a \cos[m \operatorname{ArcTan}[x]] + b \cos[n \operatorname{ArcTan}[x]])^p}{1+x^2}, x, \tan[c+dx] \right] \partial_x \tan[c+dx]$$

■ **Note:** If $\frac{m-1}{2} \in \mathbb{Z}$, then $\sin[m \operatorname{ArcTan}[x]]$ equals $\frac{x}{\sqrt{1+x^2}}$ times a rational functions in x^2 .

■ **Note:** If $\frac{m-1}{2} \in \mathbb{Z}$, then $\cos[m \operatorname{ArcTan}[x]]$ equals $\frac{1}{\sqrt{1+x^2}}$ times a rational functions in x^2 .

■ **Rule:** If $\frac{p}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$, then

$$\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx \rightarrow \frac{1}{d} \operatorname{Subst} \left[\int \frac{(a \sin[m \operatorname{ArcTan}[x]] + b \sin[n \operatorname{ArcTan}[x]])^p}{1+x^2} dx, x, \tan[c+dx] \right]$$

$$\int (a \cos[m(c+dx)] + b \cos[n(c+dx)])^p dx \rightarrow \frac{1}{d} \operatorname{Subst} \left[\int \frac{(a \cos[m \operatorname{ArcTan}[x]] + b \cos[n \operatorname{ArcTan}[x]])^p}{1+x^2} dx, x, \tan[c+dx] \right]$$

■ **Program code:**

```
Int[(a_.*sin[m_.*(c_+d_.*x_)]+b_.*sin[n_.*(c_+d_.*x_)])^p_,x_Symbol] :=
  1/d * Subst[Int[Simplify[TrigExpand[a*Sin[m*ArcTan[x]]+b*Sin[n*ArcTan[x]]]^p/(1+x^2),x],x,Tan[c+d*x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p/2,0] && IntegerQ[(m-1)/2] && IntegerQ[(n-1)/2]
```

```
Int[(a_.*cos[m_.*(c_+d_.*x_)]+b_.*cos[n_.*(c_+d_.*x_)])^p_,x_Symbol] :=
  1/d * Subst[Int[Simplify[TrigExpand[a*Cos[m*ArcTan[x]]+b*Cos[n*ArcTan[x]]]^p/(1+x^2),x],x,Tan[c+d*x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p/2,0] && IntegerQ[(m-1)/2] && IntegerQ[(n-1)/2]
```

2: $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $\frac{p-1}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$

■ **Derivation: Integration by substitution**

■ **Basis:** If $\frac{p-1}{2} \in \mathbb{Z} \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$, then

$$(a \sin[m(c+dx)] + b \sin[n(c+dx)])^p = -\frac{1}{d} \operatorname{Subst} \left[\frac{(a \sin[m \operatorname{ArcCos}[x]] + b \sin[n \operatorname{ArcCos}[x]])^p}{\sqrt{1-x^2}}, x, \cos[c+dx] \right] \partial_x \cos[c+dx]$$

■ **Basis:** If $\frac{p-1}{2} \in \mathbb{Z} \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$, then

$$(a \cos[m(c+dx)] + b \cos[n(c+dx)])^p = \frac{1}{d} \operatorname{Subst} \left[\frac{(a \cos[m \operatorname{ArcSin}[x]] + b \cos[n \operatorname{ArcSin}[x]])^p}{\sqrt{1-x^2}}, x, \sin[c+dx] \right] \partial_x \sin[c+dx]$$

■ **Note:** If $\frac{m-1}{2} \in \mathbb{Z}$, then $\sin[m \operatorname{ArcCos}[x]]$ equals $\sqrt{1-x^2}$ times a $\frac{m-1}{2}$ degree polynomial in x^2 .

■ **Note:** If $\frac{m-1}{2} \in \mathbb{Z}$, then $\cos[m \operatorname{ArcSin}[x]]$ equals $\sqrt{1-x^2}$ times a $\frac{m-1}{2}$ degree polynomial in x^2 .

■ **Rule:** If $\frac{p-1}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$, then

$$\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx \rightarrow -\frac{1}{d} \operatorname{Subst} \left[\int \frac{(a \sin[m \operatorname{ArcCos}[x]] + b \sin[n \operatorname{ArcCos}[x]])^p}{\sqrt{1-x^2}} dx, x, \cos[c+dx] \right]$$

$$\int (a \cos[m(c+dx)] + b \cos[n(c+dx)])^p dx \rightarrow \frac{1}{d} \operatorname{Subst} \left[\int \frac{(a \cos[m \operatorname{ArcSin}[x]] + b \cos[n \operatorname{ArcSin}[x]])^p}{\sqrt{1-x^2}} dx, x, \sin[c+dx] \right]$$

■ **Program code:**

```
Int[(a_.*sin[m_.*(c_+d_.*x_)]+b_.*sin[n_.*(c_+d_.*x_)])^p_,x_Symbol] :=
  -1/d * Subst[Int[Simplify[TrigExpand[a*Sin[m*ArcCos[x]]+b*Sin[n*ArcCos[x]]]]^p/Sqrt[1-x^2],x],x,Cos[c+d*x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[(p-1)/2,0] && IntegerQ[(m-1)/2] && IntegerQ[(n-1)/2]
```

```
Int[(a_.*cos[m_.*(c_+d_.*x_)]+b_.*cos[n_.*(c_+d_.*x_)])^p_,x_Symbol] :=
  1/d * Subst[Int[Simplify[TrigExpand[a*Cos[m*ArcSin[x]]+b*Cos[n*ArcSin[x]]]]^p/Sqrt[1-x^2],x],x,Sin[c+d*x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[(p-1)/2,0] && IntegerQ[(m-1)/2] && IntegerQ[(n-1)/2]
```

3: $\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$

■ **Derivation: Integration by substitution**

■ **Basis: If $m \in \mathbb{Z} \wedge n \in \mathbb{Z}$, then**

$$F[\sin[m(c+dx)], \sin[n(c+dx)]] = \frac{2}{d} \operatorname{Subst}\left[\frac{F[\sin[2m \operatorname{ArcTan}[x]], \sin[2n \operatorname{ArcTan}[x]]}{1+x^2}, x, \tan\left[\frac{1}{2}(c+dx)\right]\right] \partial_x \tan\left[\frac{1}{2}(c+dx)\right]$$

■ **Basis: If $m \in \mathbb{Z} \wedge n \in \mathbb{Z}$, then**

$$F[\cos[m(c+dx)], \cos[n(c+dx)]] = -\frac{2}{d} \operatorname{Subst}\left[\frac{F[\cos[2m \operatorname{ArcCot}[x]], \cos[2n \operatorname{ArcCot}[x]]}{1+x^2}, x, \cot\left[\frac{1}{2}(c+dx)\right]\right] \partial_x \cot\left[\frac{1}{2}(c+dx)\right]$$

■ **Note: If $m \in \mathbb{Z}$, then $\sin[2m \operatorname{ArcTan}[x]]$ equals x times a rational functions in x^2 .**

■ **Note: If $m \in \mathbb{Z}$, then $\cos[2m \operatorname{ArcCot}[x]]$ equals a rational functions in x^2 .**

■ **Rule: If $p \in \mathbb{Z}^- \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$, then**

$$\int (a \sin[m(c+dx)] + b \sin[n(c+dx)])^p dx \rightarrow \frac{2}{d} \operatorname{Subst}\left[\int \frac{(a \sin[2m \operatorname{ArcTan}[x]] + b \sin[2n \operatorname{ArcTan}[x]])^p}{1+x^2} dx, x, \tan\left[\frac{1}{2}(c+dx)\right]\right]$$

■ **Program code:**

```
Int[(a_.*sin[m_.*(c_+d_.*x_)]+b_.*sin[n_.*(c_+d_.*x_)])^p_,x_Symbol] :=
  2/d * Subst[Int[Simplify[TrigExpand[a*Sin[2*m*ArcTan[x]]+b*Sin[2*n*ArcTan[x]]]^p/(1+x^2),x],x,Tan[1/2*(c+d*x)]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IntegerQ[m/2] && IntegerQ[(n-1)/2]
```

```
Int[(a_.*cos[m_.*(c_+d_.*x_)]+b_.*cos[n_.*(c_+d_.*x_)])^p_,x_Symbol] :=
  -2/d * Subst[Int[Simplify[TrigExpand[a*Cos[2*m*ArcCot[x]]+b*Cos[2*n*ArcCot[x]]]^p/(1+x^2),x],x,Cot[1/2*(c+d*x)]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IntegerQ[m/2] && IntegerQ[(n-1)/2]
```

Rules for integrands of the form $(a \sin[m(c+dx)] + b \cos[n(c+dx)])^p$

1: $\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$

■ **Derivation: Integration by substitution**

■ **Basis:** If $\frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$, then

$$F[\sin[m(c+dx)], \cos[n(c+dx)]] = \frac{1}{d} \operatorname{Subst}\left[\frac{F[\sin[m \operatorname{ArcTan}[x]], \cos[n \operatorname{ArcTan}[x]]]}{1+x^2}, x, \tan[c+dx]\right] \partial_x \tan[c+dx]$$

■ **Note:** If $\frac{m}{2} \in \mathbb{Z}$, then $\sin[m \operatorname{ArcTan}[x]]$ equals x times a rational functions in x^2 .

■ **Note:** If $\frac{m}{2} \in \mathbb{Z}$, then $\cos[m \operatorname{ArcTan}[x]]$ equals a rational functions in x^2 .

■ **Rule:** If $p \in \mathbb{Z}^- \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n}{2} \in \mathbb{Z}$, then

$$\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx \rightarrow \frac{1}{d} \operatorname{Subst}\left[\int \frac{(a \sin[m \operatorname{ArcTan}[x]] + b \cos[n \operatorname{ArcTan}[x]])^p}{1+x^2} dx, x, \tan[c+dx]\right]$$

■ **Program code:**

```
Int[(a_.*sin[m_.*(c_+d_.*x_)]+b_.*cos[n_.*(c_+d_.*x_)])^p_,x_Symbol] :=
  1/d * Subst[Int[Simplify[TrigExpand[a*Sin[m*ArcTan[x]]+b*Cos[n*ArcTan[x]]]^p/(1+x^2),x],x,Tan[c+d*x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IntegerQ[m/2] && IntegerQ[n/2]
```

2: $\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx$ when $\frac{p-1}{2} \in \mathbb{Z}^- \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$

■ **Derivation: Integration by substitution**

■ **Basis:** If $\frac{p-1}{2} \in \mathbb{Z} \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$, then

$$(a \sin[m(c+dx)] + b \cos[n(c+dx)])^p = \frac{1}{d} \operatorname{Subst} \left[\frac{(a \sin[m \operatorname{ArcSin}[x]] + b \cos[n \operatorname{ArcSin}[x]])^p}{\sqrt{1-x^2}}, x, \sin[c+dx] \right] \partial_x \sin[c+dx]$$

■ **Note:** If $\frac{m}{2} \in \mathbb{Z}$, then $\sin[m \operatorname{ArcSin}[x]]$ equals $x \sqrt{1-x^2}$ times a $\frac{m-2}{2}$ degree polynomial in x^2 .

■ **Note:** If $\frac{m-1}{2} \in \mathbb{Z}$, then $\cos[m \operatorname{ArcSin}[x]]$ equals $\sqrt{1-x^2}$ times a $\frac{m-1}{2}$ degree polynomial in x^2 .

■ **Rule:** If $\frac{p-1}{2} \in \mathbb{Z}^- \bigwedge \frac{m}{2} \in \mathbb{Z} \bigwedge \frac{n-1}{2} \in \mathbb{Z}$, then

$$\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx \rightarrow \frac{1}{d} \operatorname{Subst} \left[\int \frac{(a \sin[m \operatorname{ArcSin}[x]] + b \cos[n \operatorname{ArcSin}[x]])^p}{\sqrt{1-x^2}} dx, x, \sin[c+dx] \right]$$

■ **Program code:**

```
Int[(a_.*sin[m_.*(c_.+d_.*x_)]+b_.*cos[n_.*(c_.+d_.*x_)])^p_,x_Symbol] :=
  1/d * Subst[Int[Simplify[TrigExpand[a*Sin[m*ArcSin[x]]+b*Cos[n*ArcSin[x]]]^p/Sqrt[1-x^2],x],x,Sin[c+d*x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[(p-1)/2,0] && IntegerQ[m/2] && IntegerQ[(n-1)/2]
```

3: $\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx$ when $p \in \mathbb{Z}^- \wedge m \in \mathbb{Z} \wedge n \in \mathbb{Z}$

■ **Derivation: Integration by substitution**

■ **Basis: If $m \in \mathbb{Z} \wedge n \in \mathbb{Z}$, then**

$$F[\sin[m(c+dx)], \cos[n(c+dx)]] = \frac{2}{d} \operatorname{Subst}\left[\frac{F[\sin[2m \operatorname{ArcTan}[x]], \cos[2n \operatorname{ArcTan}[x]]}{1+x^2}, x, \tan\left[\frac{1}{2}(c+dx)\right]\right] \partial_x \tan\left[\frac{1}{2}(c+dx)\right]$$

■ **Note: If $m \in \mathbb{Z}$, then $\sin[2m \operatorname{ArcTan}[x]]$ equals x times a rational functions in x^2 .**

■ **Note: If $m \in \mathbb{Z}$, then $\cos[2m \operatorname{ArcTan}[x]]$ equals a rational functions in x^2 .**

■ **Rule: If $p \in \mathbb{Z}^- \wedge m \in \mathbb{Z} \wedge n \in \mathbb{Z}$, then**

$$\int (a \sin[m(c+dx)] + b \cos[n(c+dx)])^p dx \rightarrow \frac{2}{d} \operatorname{Subst}\left[\int \frac{(a \sin[2m \operatorname{ArcTan}[x]] + b \cos[2n \operatorname{ArcTan}[x]])^p}{1+x^2} dx, x, \tan\left[\frac{1}{2}(c+dx)\right]\right]$$

■ **Program code:**

```
Int[(a_.*sin[m_.*(c_.+d_.*x_)]+b_.*cos[n_.*(c_.+d_.*x_)])^p_,x_Symbol] :=
  2/d * Subst[Int[Simplify[TrigExpand[a*Sin[2*m*ArcTan[x]]+b*Cos[2*n*ArcTan[x]]]^p/(1+x^2),x],x,Tan[1/2*(c+d*x)]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IntegerQ[m] && IntegerQ[n]
```